We apply *OreModules* to a differential time-delay system which describes an electric transmission line.

See D. Salamon, *Control and Observation of Neutral Systems*, Pitman, 1984, and also H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995.

```
> with(Ore_algebra):
> with(OreModules):
```

We define the Ore algebra *Alg*, where *Dt* acts as differentiation w.r.t. t and δ acts as shift operator. Note that all constants a_0 to a_5, and b_0 appearing in the system have to be declared in the definition of the Ore algebra:

```
> Alg := DefineOreAlgebra(diff=[Dt,t], dual_shift=[delta,s], polynom=[t,s],
> comm=[a[0],a[1],a[2],a[3],a[4],a[5],b[0]], shift_action=[delta,t]):
```

Next, we define the system with associated *Alg*-module M, and due to constant coefficients, we are actually going to work over a commutative polynomial ring with the indeterminates Dt and δ.

```
> R := evalm([[Dt+a[0], -(a[4]*Dt+a[0])*delta, -a[0], 0, -b[0]*Dt],
> [-delta*(a[5]*Dt+a[1]), Dt+a[1], 0, a[1], 0],
> [a[2], -a[2]*a[4]*delta, Dt, 0, -a[2]*b[0]],
> [a[3]*a[5]*delta, -a[3], 0, Dt, 0]]);
```

$$R := \begin{bmatrix} Dt + a_0 & -(a_4\,Dt + a_0)\,\delta & -a_0 & 0 & -b_0\,Dt \\ -\delta\,(a_5\,Dt + a_1) & Dt + a_1 & 0 & a_1 & 0 \\ a_2 & -a_2\,a_4\,\delta & Dt & 0 & -a_2\,b_0 \\ a_3\,a_5\,\delta & -a_3 & 0 & Dt & 0 \end{bmatrix}$$

Let us denote the *adjoint* of R by *R_adj*.

```
> R_adj := Involution(R, Alg):
> st := time(): Ext1 := Exti(R_adj, Alg, 1): time() - st; Ext1[1];
```

$$4.480$$

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}$$

Since *Ext1*[1] is the identity matrix, we see that the torsion submodule $\mathrm{t}(M)$ of the module M, which is associated with R, is the zero module. Hence, the electric transmission line is controllable and parametrizable, and we have a parametrization of the system in *Ext1*[3]:

```
> map(collect, Ext1[3], Dt);
```

$$\Big[-b_0\,Dt^4 - a_1\,b_0\,Dt^3 + (-b_0\,a_3\,a_1 - a_0\,a_2\,b_0)\,Dt^2 - a_0\,a_1\,a_2\,b_0\,Dt - a_0\,a_2\,b_0\,a_3\,a_1\Big]$$

$$\Big[-b_0\,\delta\,a_5\,Dt^4 - b_0\,\delta\,a_1\,Dt^3 + (-a_0\,a_2\,a_5\,b_0\,\delta - a_1\,b_0\,a_3\,a_5\,\delta)\,Dt^2 - a_0\,a_1\,a_2\,\delta\,b_0\,Dt$$
$$- a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\delta\Big]$$

$$\Big[(a_0\,a_2\,a_5\,\delta^2\,b_0 - a_0\,a_2\,b_0)\,Dt^2 + (a_0\,a_1\,a_2\,\delta^2\,b_0 - a_0\,a_1\,a_2\,b_0)\,Dt$$
$$+ a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\delta^2 - a_0\,a_2\,b_0\,a_3\,a_1\Big]$$

$$\Big[(a_1\,b_0\,a_3\,a_5\,\delta - b_0\,a_1\,a_3\,\delta)\,Dt^2 + a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\delta - \delta\,a_3\,b_0\,a_1\,a_2\,a_0\Big]$$

$$\Big[(-1 + a_4\,a_5\,\delta^2)\,Dt^4 + (-a_0 + a_1\,a_4\,\delta^2 + \delta^2\,a_0\,a_5 - a_1)\,Dt^3$$
$$+ (-a_3\,a_1 + a_1\,a_5\,a_3\,a_4\,\delta^2 + a_0\,a_1\,\delta^2 + \delta^2\,a_0\,a_5\,a_2\,a_4 - a_0\,a_2 - a_0\,a_1)\,Dt^2$$
$$+ (a_0\,a_1\,a_2\,a_4\,\delta^2 - a_0\,a_3\,a_1 - a_0\,a_2\,a_1 + \delta^2\,a_5\,a_0\,a_3\,a_1)\,Dt + a_0\,a_1\,a_2\,a_5\,a_3\,a_4\,\delta^2$$
$$- a_1\,a_3\,a_0\,a_2\Big]$$

The same parametrization can be obtained by using *Parametrization*. The result involves one free function ξ_1:

```
>  Parametrization(R, Alg);
```

$$\Big[-b_0\,(\mathrm{D}^{(4)})(\xi_1)(t) - b_0\,a_3\,a_1\,\%1 - a_1\,b_0\,(\mathrm{D}^{(3)})(\xi_1)(t) - a_0\,a_1\,a_2\,b_0\,\mathrm{D}(\xi_1)(t)$$
$$- a_0\,a_2\,b_0\,a_3\,a_1\,\xi_1(t) - a_0\,a_2\,b_0\,\%1\Big]$$

$$\Big[-b_0\,a_5\,a_0\,a_2\,\%3 - a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\xi_1(t-1) - b_0\,a_1\,a_3\,a_5\,\%3$$
$$- a_1\,b_0\,(\mathrm{D}^{(3)})(\xi_1)(t-1) - b_0\,a_5\,(\mathrm{D}^{(4)})(\xi_1)(t-1) - a_0\,a_1\,a_2\,b_0\,\mathrm{D}(\xi_1)(t-1)\Big]$$

$$\Big[b_0\,a_5\,a_0\,a_2\,\%2 + a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\xi_1(t-2) - a_0\,a_1\,a_2\,b_0\,\mathrm{D}(\xi_1)(t)$$
$$- a_0\,a_2\,b_0\,a_3\,a_1\,\xi_1(t) + a_0\,a_1\,a_2\,b_0\,\mathrm{D}(\xi_1)(t-2) - a_0\,a_2\,b_0\,\%1\Big]$$

$$\Big[a_0\,a_2\,a_1\,b_0\,a_3\,a_5\,\xi_1(t-1) + b_0\,a_1\,a_3\,a_5\,\%3 - a_0\,a_2\,b_0\,a_3\,a_1\,\xi_1(t-1) - b_0\,a_3\,a_1\,\%3\Big]$$

$$\Big[-a_0\,(\mathrm{D}^{(3)})(\xi_1)(t) + a_0\,a_2\,a_5\,a_4\,\%2 + a_0\,a_1\,a_2\,a_5\,a_3\,a_4\,\xi_1(t-2) + a_3\,a_5\,a_4\,a_1\,\%2$$
$$+ a_1\,a_4\,(\mathrm{D}^{(3)})(\xi_1)(t-2) + a_4\,a_5\,(\mathrm{D}^{(4)})(\xi_1)(t-2) - a_3\,a_1\,\%1 - a_0\,a_2\,\%1$$
$$+ a_0\,a_1\,a_2\,a_4\,\mathrm{D}(\xi_1)(t-2) + a_0\,a_1\,a_3\,a_5\,\mathrm{D}(\xi_1)(t-2) + a_0\,a_5\,(\mathrm{D}^{(3)})(\xi_1)(t-2)$$
$$- a_1\,a_3\,a_0\,a_2\,\xi_1(t) - a_0\,a_2\,a_1\,\mathrm{D}(\xi_1)(t) + a_0\,a_1\,\%2 - a_0\,a_3\,a_1\,\mathrm{D}(\xi_1)(t) - (\mathrm{D}^{(4)})(\xi_1)(t)$$
$$- a_1\,(\mathrm{D}^{(3)})(\xi_1)(t) - a_0\,a_1\,\%1\Big]$$

$$\%1 := (\mathrm{D}^{(2)})(\xi_1)(t)$$
$$\%2 := (\mathrm{D}^{(2)})(\xi_1)(t-2)$$
$$\%3 := (\mathrm{D}^{(2)})(\xi_1)(t-1)$$

We compute ext^2 of *R_adj*:

```
>  st := time():  Ext2 := Exti(R_adj, Alg, 2):  time() - st;
>  map(collect, Ext2[1], Dt);
```

$$3.610$$

$$\left[(-\delta\, a_1{}^2\, a_5\, a_3 + \delta\, a_3\, a_1{}^2 - \delta\, a_2\, a_1\, a_0 + \delta\, a_5\, a_0\, a_2\, a_1)\, Dt - \delta\, a_1{}^2\, a_5\, a_3{}^2 + 2\, a_0\, a_2\, a_1\, a_3\, a_5\, \delta + \delta^3\, a_1{}^2\, a_2\, a_0 - 2\, a_5{}^2\, a_0\, a_2\, a_1\, a_3\, \delta^3 + a_5{}^2\, a_0{}^2\, a_2{}^2\, \delta^3 + a_1{}^2\, a_5{}^2\, a_3{}^2\, \delta^3 - a_1{}^2\, \delta\, a_2\, a_0 - a_5\, a_0{}^2\, a_2{}^2\, \delta\right]$$
$$\left[-Dt^2 + (-a_1 + \delta^2\, a_1)\, Dt - a_3\, a_1 - a_5\, \delta^2\, a_2\, a_0 + a_1\, a_5\, a_3\, \delta^2\right]$$
$$\left[\delta\, a_2\, a_0 + \delta\, Dt^2\right]$$
$$\left[a_1\, Dt^3 + (a_1{}^2 + a_5\, a_0\, a_2 - a_1\, a_5\, a_3)\, Dt^2 + (a_3\, a_1{}^2 + a_5\, a_0\, a_1\, a_2 - a_1{}^2\, a_5\, a_3)\, Dt + \delta^2\, a_1{}^2\, a_2\, a_0 - 2\, a_5{}^2\, a_0\, a_2\, a_1\, a_3\, \delta^2 + a_5{}^2\, a_0{}^2\, a_2{}^2\, \delta^2 + a_1{}^2\, a_5{}^2\, a_3{}^2\, \delta^2 - a_1{}^2\, a_5\, a_3{}^2 + a_5\, a_0\, a_2\, a_3\, a_1\right]$$

Therefore, M is a torsion-free but not a free *Alg*-module. Hence, the *torsion-free degree* $\mathrm{i}(M)$ of M is 1. Therefore, we can find a polynomial π that contains only Dt or δ such that the system is π-free.

```
> pi := PiPolynomial(R, Alg, [delta]);
```

$$\pi := [-2\, a_0{}^2\, a_2{}^2\, a_5\, \delta^3 + a_0{}^2\, a_2{}^2\, \delta + a_3{}^2\, a_1{}^2\, \delta - 2\, \delta^5\, a_5{}^2\, a_0\, a_2\, a_1\, a_3 + \delta^5\, a_1{}^2\, a_2\, a_0 + \delta^5\, a_5{}^2\, a_0{}^2\, a_2{}^2 + \delta^5\, a_1{}^2\, a_3{}^2\, a_5{}^2 - 2\, a_3{}^2\, a_1{}^2\, a_5\, \delta^3 - 2\, a_0\, a_2\, \delta\, a_3\, a_1 + 4\, a_0\, a_2\, a_1\, a_3\, a_5\, \delta^3 + a_1{}^2\, \delta\, a_2\, a_0 - 2\, \delta^3\, a_1{}^2\, a_2\, a_0]$$

```
> factor(pi);
```

$$[\delta(-2\, \delta^2\, a_5\, a_0{}^2\, a_2{}^2 + a_0{}^2\, a_2{}^2 + a_1{}^2\, a_3{}^2 - 2\, a_5{}^2\, a_0\, a_2\, a_1\, a_3\, \delta^4 + \delta^4\, a_1{}^2\, a_2\, a_0 + a_5{}^2\, a_0{}^2\, a_2{}^2\, \delta^4 + a_1{}^2\, a_3{}^2\, a_5{}^2\, \delta^4 - 2\, \delta^2\, a_1{}^2\, a_3{}^2\, a_5 - 2\, a_3\, a_2\, a_0\, a_1 + 4\, \delta^2\, a_3\, a_5\, a_2\, a_1\, a_0 + a_1{}^2\, a_2\, a_0 - 2\, \delta^2\, a_1{}^2\, a_2\, a_0)]$$

Compare with (H. Mounier, *Propriétés structurelles des systèmes linéaires à retards: aspects théoriques et pratiques*, PhD Thesis, University of Orsay, France, 1995). We conclude that the system is π-free. Let us compute a basis of the module M over the localized ring $Alg[\pi^{-1}]$.

```
> S := LocalLeftInverse(Ext1[3], pi, Alg):
> T := map(collect, S, delta);
```

$$T := \left[-\frac{\delta\, (\delta^4\, a_5{}^2\, a_0 + (-a_5\, a_0 + a_1 - a_5\, a_1 - a_5{}^2\, a_0)\, \delta^2 + a_5\, a_0 - a_1 + a_5\, a_1)}{\%2\, (-1 + a_5)\, a_0\, b_0},\right.$$
$$-\frac{(-a_1\, a_4 + a_1\, a_5\, a_4 - a_5\, a_0)\, \delta^4 + (a_1\, a_4 - a_1\, a_5\, a_4 + a_5\, a_0 + a_0)\, \delta^2 - a_0}{\%2\, (-1 + a_5)\, a_0\, b_0},$$
$$\frac{\delta\, ((a_3\, a_5\, a_1 - a_5\, a_0\, a_2)\, \delta^2 - a_3\, a_1 + a_0\, a_2)}{\%2\, a_0\, a_2\, b_0},$$
$$-\frac{(a_5{}^2\, a_3\, a_1 - a_1{}^2 - a_5{}^2\, a_0\, a_2)\, \delta^4 + (2\, a_1{}^2 - 2\, a_3\, a_5\, a_1 + 2\, a_5\, a_0\, a_2)\, \delta^2 - a_1{}^2 - a_0\, a_2 + a_3\, a_1}{\%2\, a_3\, b_0\, a_1\, (-1 + a_5)},$$
$$\left.-\frac{\delta\, a_1\, (-1 + \delta^2)}{\%2\, a_0}\right]$$
$$\%1 := a_1{}^2\, a_2\, a_0$$
$$\%2 := (a_3{}^2\, a_5{}^2\, a_1{}^2 + \%1 + a_5{}^2\, a_0{}^2\, a_2{}^2 - 2\, a_5{}^2\, a_0\, a_2\, a_1\, a_3)\, \delta^5 + (-2\, a_1{}^2\, a_3{}^2\, a_5 - 2\, a_5\, a_0{}^2\, a_2{}^2 + 4\, a_3\, a_5\, a_2\, a_1\, a_0 - 2\, a_1{}^2\, a_2\, a_0)\, \delta^3 + (-2\, a_3\, a_2\, a_0\, a_1 + a_0{}^2\, a_2{}^2 + a_1{}^2\, a_3{}^2 + \%1)\, \delta$$

We check that the matrix T is a left-inverse of *Ext1*[3], i.e., we have $T\ Ext1[3] = 1$.

```
> simplify(evalm(T &* Ext1[3]));
```

$$\left[\ 1\ \right]$$

Hence, $z = T\ (x1\ :\ ...\ :\ x4\ :\ u)^T$ is a basis of the module asscociated with the matrix R over the localized ring $Alg[\pi^{-1}]$, which satisfies $(x1\ :\ ...\ :\ x4\ :\ u)^T = Ext1[3]\ z$.